

Dynare の使い方*

Real Arai †

平成 21 年 8 月 5 日

1 Dynare って？

Dynare is a powerful and highly customizable engine, with an intuitive front-end interface, to solve, simulate and estimate DSGE models.

2 Dynare のインストール

- 基本的に，Matlab をインストール済みであることが前提である．
- <http://www.ceprenap.cnrs.fr/juillard/mambo/index.php> の “download” から “Dynare for Matlab” バージョンをダウンロード，適当な箇所にインストールすれば OK ．
- 使用する前に，Matlab を起動させ，Dynare のフォルダまでパスを通しておくこと．これで準備完了．

3 Dynare の使い方，使用例

Dynare による数値計算

まずは，作業手順を示す．

1. Matlab を起動し，エディタで Dynare のコード（具体例は後述）を記述する．
2. Dynare を動かすコマンドを入力する．
3. 出力結果が得られる．

という流れになる．

1. Matlab を起動し，エディタで Dynare のコード（具体例は後述）を記述する．

コード入力以前に必要なことは，

- モデルに出てくる内生変数，外生変数ならびにパラメータをまとめておくこと．

*なお，Dynare コードの作成などにあたり，京都大学経済学研究科 荒戸寛樹氏（所属は作成当時）の協力を頂いた．ここに謝意を表す．

†E-mail: real@e01.mbox.media.kyoto-u.ac.jp

- 内生変数の初期値，外生変数ならびにパラメータの値を決定しておくこと．
- モデルの連立差分方程式体系を数式で表しておくこと．

である．この作業が終わったら，Matlab を起動し，エディタ画面を出して，Dynare のコードを記入する．具体的には，各節のコード具体例を参照のこと．最後にファイルの保存も忘れずに¹⁾．

2. Dynare を動かすコマンドを入力する．

次に，Matlab のコマンドラインから，`dynare FILENAME [Enter]` と入力する．ここで，FILENAME には，上で保存したファイルのファイル名を入力する（拡張子はいらぬ）．あとは Dynare が勝手に計算してくれるはずである．

3. 出力結果が得られる．

うまくいけば，計算結果が出てくるはずである．Matlab ウィンドウの“Workspace”に，各変数の出力結果が出ているはず．もちろん Matlab の描画機能を使えば，グラフ出力なども可能である．

使用例 1: RCK model with government expenditure

モデル概略

コアマクロで学習済みの，RCK モデルに政府支出が入ったモデルを考える．ただし政府支出 g の大きさは一定 ($g_t = g_1$) とする．

$$\max \sum_{t=0}^{\infty} \beta^t \ln c_t \quad (1)$$

$$\text{s.t. } y_t = c_t + k_{t+1} - (1 - \delta)k_t + g_1, \quad y_t = k_t^\alpha. \quad (2)$$

ここでは簡単化のため，人口成長率ならびに技術進歩率は 0 であると仮定しておく．

均衡における条件は以下のとおり．

$$\frac{1}{c_t} = \beta(1 - \delta + \alpha k_t^{\alpha-1}) \frac{1}{c_{t+1}} \Leftrightarrow \frac{c_{t+1}}{c_t} = \beta(1 - \delta + \alpha k_t^{\alpha-1}) \quad (3)$$

$$k_t^\alpha = c_t + k_{t+1} - (1 - \delta)k_t + g_1 \Leftrightarrow k_{t+1} - k_t = k_t^\alpha - c_t - \delta k_t - g_1 \quad (4)$$

従って，位相図を描くことによって，このモデルの動学を描写することができる．

ここで，外生変数 g_t について，「0 期時点で， T 期に g が g_1 から g_2 へと増加する」ことがアナウンスされた（そしてそれが実現する）状況を考えてみよう．このとき，家計は上記の情報を織り込んだ上で，最適なパスを決定するはずだから，

- 0 期に消費がある値にジャンプする (k はストック変数だからジャンプ不可)．
- T 期において， g が g_1 から g_2 へと増加した後の位相図における saddle path に乗る．

¹⁾このとき，拡張子が勝手に `.m` になってしまうので，Dynare ファイルの拡張子である `.mod` を必ず入力しておくこと．(例: RCK.mod みたいな感じ)

ように最適な動学は決定するはずである。

しかし、 (k, c) が具体的にどのような値を経て定常状態に向かうかは位相図からはわからない。そこで、

- 具体的な (現実的と思われる) パラメータを決定し
- 定常状態ならびに、移行経路における内生変数の値を決定する

ために数値計算が必要になる。この作業を Dynare を用いて行うことができる。具体的には、

- 定常状態における各変数の値を求める。
- 連立差分方程式体系を近似する。
- 上で近似して求めた体系をもとに、ある期にショックを加えた際の各変数のパスを求める。

ということをしてきている。

RCK モデルの Dynare コード

以下に、実際の Dynare コードの一例を挙げてみる。

```
var k c y r; % 内生変数を定義する。
varexo g; % (deterministic な) 外生変数を定義する。
parameters delta alpha beta; % パラメータを定義する。

%以下、パラメータの値を入力。
alpha = 0.36;
beta = (1.01)^(-1);
delta = 0.05;

model; % 以下、モデルを入力。
c(+1) = beta * (1 - delta + alpha*k^(alpha-1)) * c;
k = k(-1)^(alpha) - c - g + (1 - delta)*k(-1);
y = k(-1)^(alpha);
r = alpha * k(-1)^(alpha - 1);
end;

initval; % 以下、初期値を入力。
g=0.40;
c=1.51782;
k=16.4385;
r=0.06;
y=2.73974;
end;
```

```

steady; % 定常状態の値を計算する .

endval; % 最終期の値を入力 . (必要な変数のみ)
g=0.404;
end;

steady;

shocks; % (一時的な) ショックの値を入力する .
var g;
periods 1:49;
values 0.40;
end;

simul(periods=350); % 上記のショックをもとにしたシミュレーションを行う .

rplot c;
rplot k;
rplot y;

```

以上のコードを mod ファイルに保存し, Matlab 上で Dynare を実行すれば, 計算結果および動学パス $\{c_t, k_t, y_t\}_{t=0}^{350}$ のグラフが得られる²⁾ .

使用例 2: Standard RBC model

ここでは, 最も簡単な RBC モデルを用いてシミュレーションを試みる . モデルの描写は以下のとおり .

$$\max \sum_{t=0}^{\infty} \beta^t \ln c_t \quad (5)$$

$$\text{s.t. } k_{t+1} - (1 - \delta)k_t = y_t - c_t, \quad y_t = e^{A_t} k_t^\alpha, \quad A_t = \rho A_{t-1} + \epsilon_t \quad (6)$$

ここで, ϵ_t は, 平均 0, 分散 σ^2 (一定) のホワイトノイズと仮定する .

このモデルにおいて, 均衡では次の条件が全て満たされるはず .

$$\frac{1}{c_t} = E_t \left[\frac{1 - \delta + \alpha e^{A_t} k_t^{\alpha-1}}{c_{t+1}} \right] \quad (7)$$

$$k_{t+1} - (1 - \delta)k_t = e^{A_t} k_t^\alpha - c_t \quad (8)$$

$$A_t = \rho A_{t-1} + \epsilon_t \quad (9)$$

ここでは,

- 定常状態における各変数の値を計算する .

²⁾rplot c k y; とすれば, 同一平面上にグラフをプロット可能である . ただし今回の場合は, 各変数のスケールの違いのために, グラフの形状に不都合があったため, rplot 命令を 3 回使って, 別平面上に描き起こした .

- 定常状態周りで対数二次近似 (log-quadratic deviation) を行う .
- 上で近似した体系に基づいて , 各変数の transition function , 各種記述統計量 , impulse response を調べる .

という手順を Dynare で行うことにする .

RBC モデルの Dynare コード

具体的な Dynare コードは , 例えば次のように書くことができる .

```

var c k y A;
varexo e;
parameters alpha beta delta rho;

alpha=0.3;
beta=0.98;
delta=0.05;
rho=0.8;

model;
1/exp(c) = beta*(1 - delta + alpha*exp(A(+1))*exp(k)^(alpha-1))*1/exp(c(+1));
exp(y) = exp(c) + exp(k) - (1 - delta)*exp(k(-1));
exp(y) = exp(A)*exp(k(-1))^(alpha);
A = rho*A(-1) + e;
end;

initval;
c=0.381621;
k=2.07068;
y=0.621203;
e=0.00;
end;

steady;
check;

shocks;
var e = 0.05;
end;

stoch_simul( periods=350);

```

4 参考文献，その他ソフトウェア

次のサイトが特に参考になると思われるので，興味があればぜひチェックしてみるのがいいだろう．

- Dynare site

Dynare に関しては，本家のサイトをチェックするのが一番いいだろう．特に，

- Documentation — “DYNARE USER GUIDE by Tommaso Mancini Griffoli” はその他コマンドのリファレンスもあり有用．
- Examples — “J. Fernandez-Villaverde RBC models ” は，多くの種類の RBC モデルに関するコードを掲載している．
- Dynare Manual by Michel Juillard — html オンラインマニュアルができつつある．コマンドパラメータなどをさくっと調べたいときに便利である．

- 加藤涼先生の HP

Dynare とは関係ないが，独自に RBC モデルの Matlab コードを公開している．たくさんの種類のコードを公開しており，「現代マクロ経済学講義」の本と併用すればかなりの勉強ができる．

- “Solving Dynamic General Equilibrium Models Using a Second-Order Approximation to the Policy Function,”

こちらも独自に，1 次，2 次近似の為の Matlab コードなどを公開している．モデルの具体例となるコードもあり，Matlab 内での計算方法をチェックするには良いかと．

改訂

2009.08.05 例 1 のコードに不具合があったので訂正．(大阪大学経済学研究科 盛本さん，京都大学経済学研究科 天龍さんにご指摘いただきました．)

参考文献

- [1] Dynare site, <http://www.cepremap.cnrs.fr/juillard/mambo/index.php>.
- [2] Ryo Kato Web Page, <http://www.ryokato.org/>.
- [3] Matlab code of “Solving Dynamic General Equilibrium Models Using a Second-Order Approximation to the Policy Function”, Stephanie Schmitt-Grohe and Martin Uribe, http://www.econ.duke.edu/~uribe/2nd_order.htm.